

0.1 Aufbau der DMF-DSL

Im Kern des DMFs sind die Modelldateien. Sie enthalten alle Informationen und sind die Schnittstelle zwischen DMF und Entwickler*innen. Deswegen ist das Format essenziell. DMF setzt auf XML als Format für die Modelldateien und erleichtert dadurch die Bearbeitung durch grafische Editoren in Eclipse. Grafische Editoren funktionieren für das DMF nicht, da das DMF mithilfe des Language Server Protokolls nur direkt Text-Editoren unterstützt. Text-Editoren bieten eine schnellere Entwicklung, da sie direkt auf der Textbearbeitung der jeweiligen IDE aufsetzen.

Somit wurde für das DMF eine Domain Specific Language (DSL) entwickelt.

0.1.1 Die DSL und die EBNF

Die DMF-DSL besteht aus vielen Segmenten, die in einer Erweiterte Backus-Naur-Form (EBNF) dargestellt werden können. In diesem Abschnitt werden die verschiedenen Segmente und ihre Beziehung zur Abstraktion vorgestellt. Mithilfe ausgewählter EBNF-Regeln wird die Grammatik erläutert.

Die EBNF Grammatik nutzt '[' für optionale Elemente, '*' für optionale Wiederholungen (mindestens 0 Elemente) und '+' erforderliche Wiederholungen (mindestens 1 Element). Es können Elemente mit Klammern gruppiert werden. Für Tokens werden die regulären Ausdrücke angegeben und mit einem 'R' markiert.

Die Grammatik beginnt mit der Regel 'source_file':

```

<source_file>      ::= <dmf_declaration> <new_line> <model_declaration> <new_line> [ <import_block> ]
                    <model_content>

```

In ihr sind die Regeln des Headers, des Importblocks und des Modellinhalts enthalten.

Namenskonvention in der DMF Grammatik Die Grammatik des DMFs nutzt eine Namenskonvention, um den Zweck einer Regel zu dokumentieren.

Ein Block beschreibt immer die Regeln, die ein Element umschließen.

Eine Content-Regel beschreibt den Inhalt in einem Block. Ein Block kann immer mehrere Content-Elemente beinhalten. Ein Content-Element schreibt auch ein modelliertes Element, z.B. ein Argument.

Eine Value-Regel beschreibt eine Regel, welche einen Wert welcher während der semantischen Verarbeitung des ASTs eingelesen wird.

Header

Jede DMF Datei beginnt mit einem Header. Dieser Header wird genutzt die Version des Formats und Metadaten über das Modell zu dokumentieren.

Listing 1: Header einer DMF-Modelldatei

```
1 dmf 1.0.0
2 model "beispiel" version 0.0.1
```

Der Header besteht aus 2 wichtigen Grammatik-Regeln: ‘dmf_declaration’ und ‘model_declaration’. Beide bilden jeweils eine Zeile ab.

$\langle dmf_declaration \rangle ::= 'dmf' \langle version_number \rangle$

$\langle model_declaration \rangle ::= 'model' \langle string_value \rangle 'version' \langle version_number \rangle$

Importblock

Es gibt zwei Gründe für den Import von vorhandenen Modellen:

1. Um die enthaltenen Klassen in das eigene Modell zu kopieren.
2. Um die enthaltenen Klassen zu erweitern.

Um ein DMF Modell zu importieren wird die Datei und das Package das vom Modell übernommen werden soll. Es muss ein Package angegeben werden damit immer klar ist, welche Pfade schon belegt sind.

Listing 2: Import des Package de.base

```
1 import de.base from "../base.dmf"
```

Diese Importstatements werden alle in einem Importblock zusammengefasst. Sollten keine Importstatements genutzt werden, so wird der komplette Importblock aus dem AST entfernt, statt einen leeren Importblock zu enthalten.

$\langle import_block \rangle ::= \langle import_statement \rangle +$

$\langle import_statement \rangle ::= 'import' \langle package_string \rangle 'from' \langle string_value \rangle$

Modellelemente erweitern Das DMF beinhaltet die Möglichkeit Elemente zu erweitern. Dies ermöglicht Code für das ursprüngliche Modell wiederzuverwenden und trotzdem Argumente, Referenzen und Funktionen hinzuzufügen.

Eine Erweiterung im DMF muss generell explizit gekennzeichnet sein. Dafür muss zuerst das Package, indem der Entwickler*in etwas erweitern will, importiert werden. Das importierte Package muss nun im Modell modelliert werden. Es muss bei jedem, schon im importierten Modell vorhandenen Element, das Keyword ‘expand’ verwendet werden. In den mit ‘expand’ gekennzeichneten Elementen können neue Inhalte modelliert werden.

Modellinhalt

Der Modellinhalt enthält alle Elemente aus der Abstraktion des *DMFs*. Die Elemente werden in den Packages organisiert. Deshalb enthält die Regeln für den Modellinhalt auch die Regeln für Packages.

$\langle model_content \rangle ::= \langle package_content \rangle^+$

Override Die Abstraktion des *DMFs* kann bei manchen sprachspezifischen Anforderungen an ihre Grenzen stoßen. Deshalb gibt es in der *DMF DSL* das Konzept des Overrides. An jedem PackageElement und jedem Element innerhalb des PackageElements kann ein Override-Block hinzugefügt werden.

Dieser Block beinhaltet eine Section für einen Generat. In dieser Sektion können Werte überschrieben werden und sprachspezifische Werte hinzugefügt werden.

$\langle override_block \rangle ::= 'override' \{ (\langle java_override \rangle | \langle typescript_override \rangle)^* \}$

$\langle java_override \rangle ::= 'java' \{ (\langle java_annotation \rangle | \langle java_extends \rangle | \langle java_implements \rangle | \langle java_class \rangle | \langle java_name \rangle | \langle java_type \rangle | \langle java_doc \rangle)^* \}$

$\langle java_annotation \rangle ::= 'annotations' \langle stringValue \rangle$

Java Override Option	Funktion
Annotations	Fügt den Text als Annotation zu dem jeweiligen Element hinzu. Java-Annotations gehören nicht zur <i>DMF</i> Abstraktion, sind jedoch für viele Frameworks wichtig.
Extends	Überschreibt die Oberklasse.
Implements	Überschreibt die implementierten Interfaces.
Class	Überschreibt den Klassennamen.
Name	Überschreibt den Namen des Elements (z.B. Variablenname).
Type	Überschreibt den Typen des Elements (z.B. Variablentyp).
JavaDoc	Überschreibt den Kommentar.

Packages Innerhalb eines Package können weitere Packages, Structs, Entities, Enums und Interfaces enthalten sein. Diese verschiedenen Elemente werden im weiteren als PackageElemente bezeichnet.

```

1 package de.beispiel {
2
3 }
```

Jedes PackageElement kann mit einem Kommentar, dem Keyword 'expand' und einem Override-Block versehen werden:

```

1 // Das ist ein Packages mit dem Pfad "de.beispiel"
2 expand package de.beispiel {
3
4 }
5 override {
6 }

```

Die Regeln für ein Package folgen der Namenskonvention. Der 'package_string' beinhaltet Regeln für die Separation der Packages mit Punkten.

$$\begin{aligned}
 \langle package_content \rangle &::= [\langle comment_block \rangle] [\text{'expand'}] \langle package_block \rangle [\langle override_block \rangle] \\
 &\quad | [\langle comment_block \rangle] [\text{'expand'}] \langle struct_block \rangle [\langle override_block \rangle] \\
 &\quad | [\langle comment_block \rangle] [\text{'expand'}] \langle enum_block \rangle [\langle override_block \rangle] \\
 &\quad | [\langle comment_block \rangle] [\text{'expand'}] \langle entity_block \rangle [\langle override_block \rangle] \\
 &\quad | [\langle comment_block \rangle] [\text{'expand'}] \langle interface_block \rangle [\langle override_block \rangle] \\
 \langle package_block \rangle &::= \text{'package'} \langle package_string \rangle \text{'{' } \langle package_content \rangle^* \text{'}}
 \end{aligned}$$

Structs Structs beginnen wie Packages mit ihrem identifizierenden Keyword. Die Syntax für die Abstraktion folgt der Java-Syntax.

```

1 struct Beispiel extends SuperBeispiel implements IBeispiel, IABeispiel {
2
3 }

```

Die Grammatik-Regeln für ein Struct folgen der Namenskonvention. Der Inhalt eines Structs können Argumente, Referenzen, Multi-Referenzen (Collections) oder Funktionen sein.

$$\begin{aligned}
 \langle struct_block \rangle &::= \text{'struct'} \langle identifier \rangle [\langle extends_block \rangle] [\langle implements_block \rangle] \text{'{' } \langle struct_content \rangle^* \text{'}} \\
 &\quad \text{'}} \\
 \langle struct_content \rangle &::= [\langle comment_block \rangle] \langle arg_block \rangle [\langle override_block \rangle] \\
 &\quad | [\langle comment_block \rangle] \langle ref_block \rangle [\langle override_block \rangle] \\
 &\quad | [\langle comment_block \rangle] \langle multi_block \rangle [\langle override_block \rangle] \\
 &\quad | [\langle comment_block \rangle] \langle func_block \rangle [\langle override_block \rangle]
 \end{aligned}$$

Inhalt-Elemente Die Regeln für Argumente, Referenzen, Multi-Referenzen (Collections) und Funktionen werden in den verschiedenen PackageElementen wiederverwendet. Deshalb werden sie hier zentral vorgestellt.

Argument

```

1 arg int i;

```

Beim Argument werden die verschiedenen primitiven Typen mit einer Regel zusammengefasst.

$$\langle arg_block \rangle ::= \text{'arg'} \langle primitive_type \rangle \langle identifier \rangle \text{'};'$$

Referenz

```
1 ref ..beispiel.Beispiel beispiel;
2 ref de.beispiel.Beispiel zweitesBeispiel;
```

Referenzen besitzen zwei verschiedene Arten einen Typen zu spezifizieren. Entweder kann ein Typ anhand seines relativen Pfades oder seines absoluten Pfades angegeben werden. Die Punkte zu Beginn eines relativen Pfades folgen dabei den Regeln eines relativen Dateipfades.

$\langle \text{ref_block} \rangle ::= \text{'ref' } \langle \text{reftype} \rangle \langle \text{identifier} \rangle \text{' ;'}$

$\langle \text{reftype} \rangle ::= \text{' '* } \langle \text{package_string} \rangle$

Multi-Referenz

```
1 ref Map<int, .Beispiel> beispielLookup;
2 ref Set<string> namen;
```

Multi-Referenzen werden für die Collection-Typen des *DMFs* genutzt und teilen sich das Keyword 'ref' mit den normalen Referenzen. Innerhalb von '<>' können bis zu zwei Typen angegeben werden. Nur die Map erlaubt zwei Typen. Liste und Set erlauben einen. Die Typen können Referenzen oder primitive Typen sein.

$\langle \text{multi_block} \rangle ::= \text{'ref' } \langle \text{multi_name} \rangle \text{'<' } \langle \text{primitive_type} \rangle \text{'[' } \langle \text{primitive_type} \rangle \text{'>' } \langle \text{identifier} \rangle \text{' ;'}$
 $\quad \quad \quad | \text{'ref' } \langle \text{multi_name} \rangle \text{'<' } \langle \text{reftype} \rangle \text{'[' } \langle \text{primitive_type} \rangle \text{'>' } \langle \text{identifier} \rangle \text{' ;'}$
 $\quad \quad \quad | \text{'ref' } \langle \text{multi_name} \rangle \text{'<' } \langle \text{primitive_type} \rangle \text{'[' } \langle \text{reftype} \rangle \text{'>' } \langle \text{identifier} \rangle \text{' ;'}$
 $\quad \quad \quad | \text{'ref' } \langle \text{multi_name} \rangle \text{'<' } \langle \text{reftype} \rangle \text{'[' } \langle \text{reftype} \rangle \text{'>' } \langle \text{identifier} \rangle \text{' ;'}$

Funktionen

```
1 func void displayBeispiel(.Beispiel beispiel);
2 func int add(int a, int b);
3 func void execute();
```

Funktionen bestehen aus dem Keyword 'func' dem Rückgabetypen, dem Namen und den Parametern. Der Rückgabetypp kann ein primitiver Typ, eine Referenz oder 'void' sein. 'void' bedeutet, dass kein Wert zurückgegeben wird. Die Parameter bestehen aus einem Typen, der entweder zu den primitiven Typen gehört oder eine Referenz ist, und einem Namen.

$\langle \text{func_block} \rangle ::= \text{'func' } \langle \text{reftype} \rangle \langle \text{identifier} \rangle \text{'(' } [\langle \text{param_definition} \rangle \text{'(' } \langle \text{param_definition} \rangle \text{)*}] \text{')' ;'}$
 $\quad \quad \quad | \text{'func' } \langle \text{primitive_type} \rangle \langle \text{identifier} \rangle \text{'(' } [\langle \text{param_definition} \rangle \text{'(' } \langle \text{param_definition} \rangle \text{)*}] \text{')' ;'}$
 $\quad \quad \quad | \text{'func' 'void' } \langle \text{identifier} \rangle \text{'(' } [\langle \text{param_definition} \rangle \text{'(' } \langle \text{param_definition} \rangle \text{)*}] \text{')' ;'}$

$\langle \text{param_definition} \rangle ::= \langle \text{reftype} \rangle \langle \text{identifier} \rangle$
 $\quad \quad \quad | \langle \text{primitive_type} \rangle \langle \text{identifier} \rangle$

Entity Entities sind im Aufbau sehr ähnlich zu Structs. Sie unterscheiden sich im Keyword und im Identifier-Statement.

```
1 entity BeispielEntity extends AndereEntity implements IBeispiel, IABeispiel {
2     arg int i;
3     arg double d;
4     identifier(i,d);
5 }
```

Besonders bei der 'entity_block'-Regel ist, dass sie direkt die 'struct_content'-Regel verwendet. Das Identifier-Statement schließt immer den Inhalt einer Entity ab. Hierbei ist zu beachten, dass 'identifier' als Keyword und als Regelname verwendet wird.

$$\langle \text{entity_block} \rangle ::= \text{'entity'} \langle \text{identifier} \rangle [\langle \text{extends_block} \rangle] [\langle \text{implements_block} \rangle] \text{'{' } \langle \text{struct_content} \rangle^* \langle \text{identifier_statement} \rangle \text{'}'}$$
$$\langle \text{identifier_statement} \rangle ::= \text{'identifier'} \text{'(' } \langle \text{identifier} \rangle \text{'(' } \langle \text{identifier} \rangle^* \text{' ')} \text{'};'$$

Interface Der Aufbau eines Interfaces ist simpler im Vergleich zu einem Struct. Es entfällt der Extends-Block und alle Inhalte bis auf Funktionen.

```
1 interface IBeispiel implements IAnderesBeispiel {
2
3 }
```

Da nur Funktionen in einem Interface vorkommen können, besitzt die Content-Regel nur eine Variante.

$$\langle \text{interface_block} \rangle ::= \text{'interface'} \langle \text{identifier} \rangle [\langle \text{implements_block} \rangle] \text{'{' } \langle \text{interface_content} \rangle^* \text{'}'}$$
$$\langle \text{interface_content} \rangle ::= [\langle \text{comment_block} \rangle] \langle \text{func_block} \rangle [\langle \text{override_block} \rangle]$$

Enum Enum beinhalten ein neues Element: die Konstante.

Konstanten sind die Einträge eines Enums und bilden mit Argumenten den Inhalt eines Enums.

```
1 enum EBeispiel {
2     arg int i;
3
4     KONSTANTE(_,1);
5 }
```

Konstanten beinhalten immer einen Index. Dieser Index wird in Datenbanktabellen hinterlegt. Es müssen zusätzlich für jedes Argument ein passender Wert angegeben werden. Damit das DMF den Index automatisch berechnet kann ein '_' verwendet werden. Die Regel 'enum_index' bildet diese Besonderheit ab.

Die Regel 'primitive_value' vereinigt alle Value-Regeln.

```

<enum_block>      ::= 'enum' <identifier> '{' <enum_content>* '}'
<enum_content>    ::= [<comment_block>] <arg_block> [<override_block>]
                    | [<comment_block>] <enum_constant> [<override_block>]
<enum_constant>   ::= <identifier> '(' <enum_index> (',' <primitive_value>)* ')' ','
<enum_index>      ::= '_' | <integerValue>

```

0.1.2 Beispieldatei

```

1  dmf 1.0.0
2  model "beispiel" version 0.0.1
3
4  import de.base from "../base.dmf"
5
6  expand package de.base {
7      expand interface IBeispiel {
8          func string printBeispielMarkdown();
9      }
10 }
11
12 package de.beispiel {
13     struct Beispiel implements ..base.IBeispiel {
14         arg int i;
15         ref .BeispielTyp typ;
16     }
17
18     entity Aufgabe {
19         ref .Beispiel beispiel;
20         arg string frage;
21         arg string antwort;
22         arg int id;
23         identifier(id);
24     }
25
26     enum BeispielTyp {
27         CODE(_);
28         TEXT(_);
29     }
30 }

```

Dieses Beispiel zeigt wie man mithilfe des *DMFs* eine Struktur modelliert. Diese Struktur beinhaltet Aufgaben, die Benutzer*innen beantworten sollen. Als Hilfestellung gibt es ein Beispiel. Zur korrekten Darstellung wird am Beispiel der *BeispielTyp* referenziert. Ein Beispiel nutzt ein Interface aus einem anderen Modell. Dieses Interface wird mit einer neuen Methode erweitert, um das Beispiel auch in Markdown zu rendern.