0.1 EBNF Grammatik für DMF

⟨*source_file*⟩ ::= ⟨*dmf_declaration*⟩ ⟨*new_line*⟩ ⟨*model_declaration*⟩ ⟨*new_line*⟩ [⟨*import_block*⟩] ⟨*model_content*⟩

⟨*dmf_declaration*⟩ ::= 'dmf' ⟨*version_number*⟩

⟨*model_declaration*⟩ ::= 'model' ⟨*string_value*⟩ 'version' ⟨*version_number*⟩

⟨*import_block*⟩ ::= ⟨*import_statement*⟩+

⟨*import_statement*⟩ ::= 'import' ⟨*package_string*⟩ 'from' ⟨*string_value*⟩ ⟨*new_line*⟩

⟨*model_content*⟩ ::= ⟨*package_content*⟩+

⟨*package_content*⟩ ::= [⟨*comment_block*⟩] ['expand'] ⟨*package_block*⟩ [⟨*override_block*⟩]
| [⟨*comment_block*⟩] ['expand'] ⟨*struct_block*⟩ [⟨*override_block*⟩]
| [⟨*comment_block*⟩] ['expand'] ⟨*enum_block*⟩ [⟨*override_block*⟩]
| [⟨*comment_block*⟩] ['expand'] ⟨*entity_block*⟩ [⟨*override_block*⟩]
| [⟨*comment_block*⟩] ['expand'] ⟨*interface_block*⟩ [⟨*override_block*⟩]

⟨*comment_block*⟩ ::= ⟨*comment*⟩+

⟨*comment*⟩ ::= R'//.*\n'

⟨*package_block*⟩ ::= 'package' '{' ⟨*package_content*⟩* '}'

⟨*struct_block*⟩ ::= 'struct' ⟨*identifier*⟩ [⟨*extends_block*⟩] [⟨*implements_block*⟩] '{' ⟨*struct_content*⟩* '}'

⟨*extends_block*⟩ ::= 'extends' ⟨*reftype*⟩

⟨*implements_block*⟩ ::= 'implements' ⟨*reftype*⟩ (',' ⟨*reftype*⟩)+

⟨*struct_content*⟩ ::= [⟨*comment_block*⟩] ⟨*arg_block*⟩ [⟨*override_block*⟩]
| [⟨*comment_block*⟩] ⟨*ref_block*⟩ [⟨*override_block*⟩]
| [⟨*comment_block*⟩] ⟨*multi_block*⟩ [⟨*override_block*⟩]
| [⟨*comment_block*⟩] ⟨*func_block*⟩ [⟨*override_block*⟩]

⟨*arg_block*⟩ ::= 'arg' ⟨*primitive_type*⟩ ⟨*identifier*⟩ ';'

⟨*ref_block*⟩ ::= 'ref' ⟨*reftype*⟩ ⟨*identifier*⟩ ';'

⟨*multi_block*⟩ ::= 'ref' ⟨*multi_name*⟩ '<' ⟨*primitive_type*⟩ [',' ⟨*primitive_type*⟩] '>' ⟨*identifier*⟩ ';'
| 'ref' ⟨*multi_name*⟩ '<' ⟨*reftype*⟩ [',' ⟨*primitive_type*⟩] '>' ⟨*identifier*⟩ ';'
| 'ref' ⟨*multi_name*⟩ '<' ⟨*primitive_type*⟩ [',' ⟨*reftype*⟩] '>' ⟨*identifier*⟩ ';'
| 'ref' ⟨*multi_name*⟩ '<' ⟨*reftype*⟩ [',' ⟨*reftype*⟩] '>' ⟨*identifier*⟩ ';'

⟨*func_block*⟩ ::= 'func' ⟨*reftype*⟩ ⟨*identifier*⟩ '(' [⟨*param_definition*⟩ (',' ⟨*param_definition*⟩)*] ')' ';'
| 'func' ⟨*primitive_type*⟩ ⟨*identifier*⟩ '(' [⟨*param_definition*⟩ (',' ⟨*param_definition*⟩)*] ')' ';'
| 'func' 'void' ⟨*identifier*⟩ '(' [⟨*param_definition*⟩ (',' ⟨*param_definition*⟩)*] ')' ';'

⟨*param_definition*⟩ ::= ⟨*reftype*⟩ ⟨*identifier*⟩
| ⟨*primitive_type*⟩ ⟨*identifier*⟩

⟨*enum_block*⟩ ::= 'enum' ⟨*identifier*⟩ '{' ⟨*enum_content*⟩* '}'

⟨*enum_content*⟩ ::= [⟨*comment_block*⟩] ⟨*arg_block*⟩ [⟨*override_block*⟩]
| [⟨*comment_block*⟩] ⟨*enum_constant*⟩ [⟨*override_block*⟩]

⟨*enum_constant*⟩ ::= ⟨*identifier*⟩ '(' ⟨*enum_index*⟩ (',' ⟨*primitive_value*⟩)* ')' ';'

⟨*enum_index*⟩ ::= '_' | ⟨*integerValue*⟩

⟨*entity_block*⟩ ::= 'entity' ⟨*identifier*⟩ [⟨*extends_block*⟩] [⟨*implements_block*⟩] '{' ⟨*struct_content*⟩* ⟨*identifier_statement*⟩ '}'

⟨*identifier_statement*⟩ ::= 'identifier' '(' ⟨*identifier*⟩ (',' ⟨*identifier*⟩)* ')' ';'

⟨*interface_block*⟩ ::= 'interface' ⟨*identifier*⟩ [⟨*implements_block*⟩] '{' ⟨*interface_content*⟩* '}'

⟨*interface_content*⟩ ::= [⟨*comment_block*⟩] ⟨*func_block*⟩ [⟨*override_block*⟩]

0.2 Beispiel

⟨*statement*⟩ ::= ⟨*ident*⟩ '`=`' ⟨*expr*⟩
| '`for`' ⟨*ident*⟩ '`=`' ⟨*expr*⟩ '`to`' ⟨*expr*⟩ '`do`' ⟨*statement*⟩
| '`{`' ⟨*stat-list*⟩ '`}`'
| ⟨*empty*⟩

⟨*stat-list*⟩ ::= ⟨*statement*⟩ '`;`' ⟨*stat-list*⟩ | ⟨*statement*⟩